


Pensieve header: Notebook for Wednesday September 16: A faster Jones program.

Warning: Hidden cell below:


The Gompf-Scharlemann-Thompson 48-crossing

GST48:




```
GST48 = PD[X[1, 15, 2, 14], X[29, 2, 30, 3], X[40, 4, 41, 3],
  X[4, 44, 5, 43], X[5, 26, 6, 27], X[95, 7, 96, 6], X[7, 1, 8, 96], X[8, 14, 9, 13],
  X[28, 9, 29, 10], X[41, 11, 42, 10], X[11, 43, 12, 42], X[12, 27, 13, 28],
  X[15, 31, 16, 30], X[61, 16, 62, 17], X[72, 17, 73, 18], X[83, 18, 84, 19],
  X[34, 20, 35, 19], X[20, 89, 21, 90], X[92, 21, 93, 22], X[22, 79, 23, 80],
  X[23, 68, 24, 69], X[24, 57, 25, 58], X[56, 25, 57, 26], X[31, 63, 32, 62],
  X[32, 74, 33, 73], X[33, 85, 34, 84], X[35, 50, 36, 51], X[81, 37, 82, 36],
  X[70, 38, 71, 37], X[59, 39, 60, 38], X[54, 39, 55, 40], X[55, 45, 56, 44],
  X[45, 59, 46, 58], X[46, 70, 47, 69], X[47, 81, 48, 80], X[91, 49, 92, 48],
  X[49, 91, 50, 90], X[82, 52, 83, 51], X[71, 53, 72, 52], X[60, 54, 61, 53],
  X[74, 63, 75, 64], X[85, 64, 86, 65], X[65, 76, 66, 77], X[66, 87, 67, 88],
  X[94, 67, 95, 68], X[86, 75, 87, 76], X[77, 88, 78, 89], X[93, 78, 94, 79]];
```

The 55-crossing Piccirillo's Knot, used to prove that "The Conway Knot is not Slice" (arXiv:1808.02923): \_\_\_\_\_



Our blackboard from September 14:



```
In[ ]:= Knot[3, 1]
```

```
KB[pd_PD] := Module[{p, t1, t2, t3, t4, B, d},
  SetAttributes[p, Orderless];
  t1 = pd /. X[i_, j_, k_, l_] -> A * p[i, j] * p[k, l] + B * p[i, l] * p[j, k];
  t2 = Expand[t1 /. PD -> Times];
  t3 = t2 //. {p[i_, j_] p[j_, k_] -> p[i, k]};
  t4 = t3 /. {p[i_, i_] -> d, p[i_, j_]^2 -> d};
  Expand[t4 /. {B -> 1/A, d -> -A^2 - 1/A^2}]
]
```

```
In[ ]:= KB[Knot[8, 17]]
```

```
In[ ]:= KBon10 = Table[
  Echo[Timing[{k, KB[Knot[10, k]}]]],
  {k, 165}
]
```

```
Plus @@ KBon10
```

```
In[ ]:= KB[GST48]
```

```

FKB[pd_PD] := Module[{p, t1, t2, t3, t4, B, d, KB, todo},
  SetAttributes[p, Orderless];
  KB = 1;
  todo = pd;
  While[Length[todo] > 0,
    x = First[todo];
    todo = DeleteCases[todo, x];
    t1 = KB (x /. X[i_, j_, k_, l_] → A * p[i, j] * p[k, l] + B * p[i, l] * p[j, k]);
    t2 = Expand[t1];
    t3 = t2 /. {p[i_, j_] * p[j_, k_] → p[i, k]};
    t4 = t3 /. {p[i_, i_] → d, p[i_, j_]^2 → d};
    KB = Expand[t4 /. {B → 1/A, d → -A^2 - 1/A^2}];
  ];
  KB
]

In[ ]:= FKB[Knot[8, 17]]

In[ ]:= Short[FKBon10 = Table[
  Timing[{k, FKB[Knot[10, k]}],
  {k, 165}
]]

In[ ]:= Plus @@ FKBon10

EFKB[pd_PD] := Module[{p, t1, t2, t3, t4, B, d, KB, todo, front, x, v},
  SetAttributes[p, Orderless];
  KB = 1;
  todo = List @@ pd;
  front = {};
  v[x_X] := Length[front ∩ (List @@ x)];
  While[Length[todo] > 0,
    x = RandomChoice[MaximalBy[todo, v]];
    todo = DeleteCases[todo, x];
    t1 = KB (x /. X[i_, j_, k_, l_] → A * p[i, j] * p[k, l] + B * p[i, l] * p[j, k]);
    t2 = Expand[t1];
    t3 = t2 /. {p[i_, j_] * p[j_, k_] → p[i, k]};
    t4 = t3 /. {p[i_, i_] → d, p[i_, j_]^2 → d};
    KB = Expand[t4 /. {B → 1/A, d → -A^2 - 1/A^2}];
    front = Complement[front ∪ (List @@ x), front ∩ (List @@ x)];
  ];
  KB
]

In[ ]:= Short[EFKBon10 = Table[
  Timing[{k, EFKB[Knot[10, k]}],
  {k, 165}
]]
Plus @@ EFKBon10

In[ ]:= Timing[EFKB[GST48]]

```